



Processing – a programming language for graphical and image based computations.

Free and available for Windows or Mac!

<http://www.processing.org/download>

Download the Windows 64 bit version. You can choose “No Donation”



Let's write some code!

Project 1: Some lines

In the Window type:

```
// Setup the Window
void setup(){
  size(400, 400); // size of window, in pixels
  background(192,64,0); // background color: R, G, B
}

// Draw a line when the mouse is pressed!
void draw(){
  if(mousePressed){ // only when the mouse is pressed!
    stroke(255); // White!

    // draw a line from 158,158 to the current
    // position of the mouse
    line(158,158, mouseX, mouseY);
  }
}
```

Press Run.

File | Save as | Desktop Sketch_1_PUNetID

Notes:

// Comments

Functions:

Colors: R, G, B

if: A decision

primitives.

mousePressed, mouseX, mouseY

Project 2: A Face

```
void setup(){
  size(400, 400); // size of window, in pixels
}



void draw(){
  background(192,64,0); // background color: R, G, B

  // yellow circle
  fill(250, 238, 8);
  ellipse(50,50,60,60);

  // black circle
  fill(0,0,0);
  ellipse(45,45,10,10);

  // black circle
  fill(0,0,0);
  ellipse(65,45,10,10);

  // thick black line
  stroke(0,0,0);
  strokeWeight(3);
  line(45, 65, 65, 65);
}
```



```
int x = 50; // a variable holds the value of x!

void setup(){
  size(400, 400); // size of window, in pixels
}

void draw(){

  background(192,64,0); // background color: R, G, B

  // yellow circle
  fill(250, 238, 8);
  ellipse(x,50,60,60);

  // black circle
  fill(0,0,0);
  ellipse(x-5,45,10,10);

  // black circle
  fill(0,0,0);
  ellipse(x+15,45,10,10);

  // thick black line
  stroke(0,0,0);
  strokeWeight(3);
  line(x-5, 65, x+15, 65);

  x = x + 1;
}
```

Change all the references to the x coordinate to use the **variable x**! Update the variable x to move one pixel at a time.

Project 4: Many Moving Faces! Functions!

What happens if background() is in face()?

What happens if `x = x + 1` is in face()?

What happens if `x = x + 5` is used?

```
int x = 50;
void setup(){
  // size of window, in pixels
  size(400, 400);
}

void draw(){
  // background color: R, G, B
  background(192,64,0);
  face(50);
  face(150);
  face(250);
  x = x + 1;
}

void face(int y){
  // yellow circle
  fill(250, 238, 8);
  ellipse(x,y,60,60);

  // black circle
  fill(0,0,0);
  ellipse(x-5,y-5,10,10);

  // black circle
  fill(0,0,0);
  ellipse(x+15,y-5,10,10);

  // thick black line
  stroke(0,0,0);
  strokeWeight(3);
  line(x-5, y+15, x+15, y+15);
}
```

Practice – you are not required to finish these questions or submit these solutions.

1. Use the quad() function to draw a kite on the screen.



2. Write function to draw a kite on the screen and use that function to place 4 kites on the screen.

3. Have the kites slowly move on the screen in any direction.

4. Have half the kites move left to right, and the other half move right to left. Hint: You may need a different function for each direction!

Tricky:

5. The code on the right will set the background to black if the user clicks in the bottom half of the screen and will set the background to white if the users clicks in the top half of the screen.

Draw a square on the screen. If the user clicks above the square change the square's color to white. If the user clicks below the square, change the square's color to black.

Very, Very, Tricky:

Draw a square on the screen. If the user clicks inside the square change the square's color to white. If the user clicks outside the square, change the square's color to black.

```
int currentColor = 0;

void setup()
{
  size(400,400);
}

void draw()
{
  background(currentColor);
}

void mouseClicked()
{
  if ( 200 <= mouseY )
  {
    currentColor = 0;
  }
  if( 200 > mouseY )
  {
    currentColor = 255;
  }
}
```